Sheel Shah, 19D070052, Expt4

Q1.

Analysis:

Vb = Vcc – Vz = 6.4. Ve = 6.4 + 0.7 = 7.1. Ie = (Vcc-Ve)/Re = 1.04mA.

Ic= Ie, as Ib = 0. Hence Iout = 1.04mA.

Code:

```
Sheel Shah 19D070052 Expt4 Current Source

.include all_model_files/zener_B.txt
.model bc557a PNP IS=10f BF=100 ISE=10.3f IKF=50m NE=1.3
+ BR=9.5 VAF=80 IKR=12m ISC=47p NC=2 VAR=10 RB=280 RE=1 RC=40
+ tr=0.3u tf=0.5n cje=12p vje=0.48 mje=0.5 cjc=6p vjc=0.7 mjc=0.33 kf=2f

** nodes:
** 1 vcc, 2 emitter, 3 base, 4/5 collector
q1 4 3 2 bc557a
v_cc 1 0 12
x_z 3 1 DI_1N4734A
r_e 1 2 4.7k
r_b 3 0 2.2k
v0 4 5 0
r_l 5 0

.dc r_l 1k 10k 1k
.control
run
plot i(v0)
.endc
.end
```

```
v(2) = 7.176954e+00
v(3) = 6.516294e+00
v(4) = 1.012825e+00
i(v0) = 1.012825e-03
```



Learnings:

1. I learned about the BJT current source, and saw that our analysis matched well.
2. R_l has an upper limit beyond which the BJT enters saturation, and I_l drops

Q2.

Analysis:

Iref = 12 – 0.7 / 10k = 1.13mA

Iout/Iref = (1/(1+2/beta))*(1 + (Vo – Vbe)/Va)

Substituting: case 1 – 1.167mA. case 2 – 1.1317mA

Code:

```
Sheel Shah 19D070052 Expt4 Current Mirror

.model bc547a NPN IS=10f BF=200 ISE=10.3f IKF=50m NE=1.3
+ BR=9.5 VAF=80 IKR=12m ISC=47p NC=2 VAR=10 RB=280 RE=1 RC=40
+ tr=0.3u tf=0.5n cje=12p vje=0.48 mje=0.5 cjc=6p vjc=0.7 mjc=0.33 kf=2f

** nodes:
** 1 vcc
** 2 collector1/base1/base2
** 3 collector2/vout
q1 2 2 0 bc547a
q2 3 2 0 bc547a
v_cc 1 0 12
v_out 3 0
r0 1 2 10k

.dc v_out 1 5 0.5
.control
run
plot i(v_out), i(v_cc)
.endc
.end
```

```
i(v_out) = -1.11923e-03
i(v_cc) = -1.13359e-03
v(2) = 6.641247e-01
```



The currents are negative as I avoided using a dummy voltage source to measure the current, and directly used the voltage sources already in the circuit.

Learnings:

The effect of V_a(early voltage) is very clearly visible as we see increasing Iout for increasing Vout. We usually assume Vbe as 0.7 and this is not very accurate.

Q3.

A:

Analysis:

Vb = 0 => Ve = -0.7. Ie = 12 – 0.7 /Re = 1.13mA.

Ic1 =Ic2 = Ie/2 = 0.565mA. Vc1 = Vc2 = 12 -IcRc = 8.158V

```
Sheel Shah 19D070052 Expt4 Diff Pair

.model bc547a NPN IS=10f BF=200 ISE=10.3f IKF=50m NE=1.3
+ BR=9.5 VAF=80 IKR=12m ISC=47p NC=2 VAR=10 RB=280 RE=1 RC=40
+ tr=0.3u tf=0.5n cje=12p vje=0.48 mje=0.5 cjc=6p vjc=0.7 mjc=0.33 kf=2f

** nodes:
** 1 emitter of both
** 2 base1, 3 base2
** 20 in1, 30 in2
** 4 collector1, 5 collector2
** 6 vee, 7 vcc
q1 4 2 1 bc547a
q2 5 3 1 bc547a

v_cc 7 0 12
v_ic1 7 74 0
r_c1 74 4 6.8k
v_ic2 7 75 0
r_c2 75 5 6.8k

v_i1 20 0 0
r_b1 2 20 1k
v_i2 30 0 0
r_b2 3 30 1k

v_ee 6 0 -12
r_e 6 1 10k

.op
.control
run
print i(v_ee), i(v_ic1), i(v_ic2)
print v(1), v(4), v(5)
.endc
.end
```

```
i(v_ee) = 1.135394e-03
i(v_ic1) = 5.630775e-04
i(v_ic2) = 5.630775e-04
v(1) = -6.46059e-01
v(4) = 8.171073e+00
v(5) = 8.171073e+00
```

The code for the following parts is very similar and hasn't been included.

B:

Analysis:
Gm = Ic/Vt = 0.0217. Single ended gain at Vo1 = -Gm*Rc/2 = -73.9 V/V. At Vo2: +73.9V/V

Simulation:



Gain Seen = 5.668392e+01
Code used:

```
plot v(20), v(4), v(5)
meas tran max_i1 max v(20)
meas tran min_i1 min v(20)
meas tran max_o1 max v(4)
meas tran min_o1 min v(4)
print (max_o1 - min_o1) / (max_i1 - min_i1)
```

C:



Learnings:

Vo1 and Vo2 are perfectly out of phase, and our theoretic formulae approximate the gain fairly well. There is however asymmetry in the final plot, and I don't know the reason behind it